

Música por computadora

Ernesto Romero y Hernani Villaseñor

Centro Multimedia 2012

SESIÓN 13

Síntesis Sustractiva

La síntesis sustractiva funciona bajo el principio de restar o moldear sonido complejo mediante filtros. Generalmente se trabaja con ruido o sonidos complejos con un amplio espectro de frecuencias los cuales son filtrados hasta obtener un timbre deseado.


```
// ejemplo con un filtro sobre ruido rosa
{BPF.ar(PinkNoise.ar(0.5), MouseX.kr(40,10000),
MouseY.kr(0.01,2.0))}.scope

// síntesis sustractiva más compleja
(
SynthDef(\sustractiva,{|rq=0.01,c_frec=#[1,2,3,4,5,6,7,8,9,10],
gate=1|
    var sen, senfilt, mix, env;
    sen=WhiteNoise.ar(1);
    senfilt=Mix(BPF.ar(sen,c_frec,rq,1));
    env=EnvGen.kr(Env.asr(0.1,1,1),gate,doneAction:2);
    Out.ar(0,senfilt*env)
    }).add
)

a=Array.exprand(10,1,1.5)
d=Synth(\sustractiva,[\c_frec,a*200])
d.set(\c_frec,a*exprand(200,1500))
d.set(\gate,0)
```

Modulación

Revisemos primero el concepto de Oscilador de Baja Frecuencia o LFO (Low Frequency Oscillator). El concepto de LFO es el de una frecuencia muy baja que sirve para modular una señal. Una oscilación puede ser vista como movimiento de vaivén, el cual puede ser usado para modular el parámetro de frecuencia o el de amplitud, en ambos casos escucharemos un sonido que se mueve en su altura o en su intensidad a la velocidad que oscila la baja frecuencia. Este comportamiento es el principio de la síntesis de Frecuencia Modulada (FM) y Amplitud Modulada (AM) respectivamente, las cuales veremos con detalle más adelante.



Gráfica que representa la modulación de amplitud.

LF-Unit Generators (Low Frequency UGens) Osciladores de baja frecuencia

SC tiene algunos UGens que están diseñados para este propósito. Estos generadores pueden ser usados para controlar o modular argumentos de la señal a partir de bajas frecuencias, aunque también pueden sonar, pero teniendo en cuenta que a altas frecuencias generan frecuencias alias.

LFPAr, LFCub, LFTri, Impulse, LFSaw, LFPulse, VarSaw, SyncSaw

LFPAr, LFCub, LFTri, LFSaw, Impulse

Argumentos: frecuencia, fase, multiplicador y suma

LFPulse, VarSaw

Argumentos: frecuencia, fase, ancho de pulso, multiplicador y suma

SyncSaw

Argumentos: syncFrec, sawFrec, multiplicador y suma

Ejemplo de un SinOsc controlado en su frecuencia por un LFO (oscilador de baja frecuencia).

```
// oscilación lenta  
{SinOsc.ar(LFPar.ar(0.2, 0, 200, 800), 0, 0.5)}.play
```

```
// oscilación rápida  
{SinOsc.ar(LFPar.ar(20, 0, 200, 800), 0, 0.5)}.play
```

Tipos de síntesis por modulación

La modulación no es otra cosa que alterar la amplitud, frecuencia o fase de un oscilador por otra señal.

Frecuencia portadora: es la frecuencia que es modulada.

Frecuencia modulante: es la frecuencia que modula.

Bandas laterales: son frecuencias resultantes de la modulación.

Modulación AM

Dodge y Jerse (90), mencionan que hay tres tipos de amplitud modulada: la “clásica” amplitud modulada (AM), modulación por anillo (Ring Modulator) y modulación de banda lateral sencilla (single-sideband). De estas tres técnicas la AM y la modulación por anillo son las que revisaremos.

Para realizar síntesis AM se necesitan dos señales: una portadora y una modulante. Llamaremos amp a un valor de amplitud que será común para ambas frecuencias. En la señal modulante amp es multiplicada por un índice de modulación que normalmente va de 0 a 1. Si su valor es 0 no habrá modulación. Si el valor es 1 la modulación será máxima. En la señal portadora amp es multiplicado por la señal moduladora. El resultado final es una AM. En la técnica de AM escuchamos la frecuencia portadora y dos bandas laterales resultantes que son la suma y la diferencia de la frecuencia portadora y la frecuencia modulante.

banda lateral 1 = F portadora + F modulante

banda lateral 2 = F portadora - F modulante

Si tenemos una frecuencia portadora de 550 y una frecuencia modulante de 150, las bandas laterales serán de 300 y 700.

```
(  
SynthDef(\am, {|frec_portadora=450, frec_modulante=2, ind_mod=1,  
amp=0.5, gate=1|  
    var portadora, modulante, envolvente;  
    modulante=SinOsc.kr(frec_modulante,0,amp*ind_mod, add:1);  
    portadora= SinOsc.ar(frec_portadora,0,amp*modulante);  
    envolvente=EnvGen.kr(Env.asr(0.1,1,1),gate,doneAction:2);  
    Out.ar(0,Pan2.ar(portadora,0,1)*envolvente)  
}).add  
)
```

```
b=Synth(\am)  
b.set(\gate,0)
```

La amplitud de las bandas laterales será de $(\text{ind_mod}/2) * \text{amp}$. Así, si la amplitud es 0.5, y el índice de modulación 1 tenemos que la amplitud de cada una de las bandas laterales es de $(1/2) * 0.5 = 0.25$. La amplitud total de la onda resultante es la suma de la amplitud de la frecuencia portadora más las amplitudes de las laterales.

```
amp portadora = 0.5  
amp lateral 1 = 0.25  
amp lateral 2 = 0.25
```

```
amp total = 1
```

Podemos comparar el resultado de la AM con portadora en 550 y modulante en 150 con tres sinusoidales con las frecuencias de la portadora y las bandas laterales y sus respectivas amplitudes.

```
~l1={SinOsc.ar(550 - 150, 0, 0.25)}.play;  
~fp={SinOsc.ar(550, 0, 0.5)}.play;  
~l2={SinOsc.ar(550 + 150, 0, 0.25)}.play;  
  
~l1.free;  
~fp.free;  
~l2.free;
```

Modulación por Anillo

La modulación por anillo o *ring modulator* es un tipo de síntesis que toma su nombre de la forma que tenían los circuitos cuando eran diseñados de manera análoga, esta síntesis funciona al multiplicar la frecuencia portadora por la frecuencia modulante. Se utiliza solo una amplitud para la señal moduladora. Es importante recalcar que solo suenan la frecuencia de las bandas laterales.

F portadora * F modulante

```
(
SynthDef(\ringmodul, {|frec_portadora=550, frec_modulante=150,
amp=1, gate=1|
    var portadora, modulante, envolvente;
    modulante=SinOsc.ar(frec_modulante, 0, amp);
    portadora= SinOsc.ar(frec_portadora, 0, modulante);
    envolvente=EnvGen.kr(Env.asr(0.1, 1, 1), gate, doneAction:2);
    Out.ar(0, portadora*envolvente)
    }).add
)
```

```
a=Synth(\ringmodul)
a.set(\gate, 0)
```

Modulación FM

La modulación FM, según Hutchins (10), fue descubierta en Stanford por John Chowing en 1973. Posteriormente la licencia fue adquirida por Yamaha para implementarla en el sintetizador DX7.

En esta técnica la frecuencia moduladora hace su trabajo de modulación sobre la frecuencia portadora, causando un tipo de vibrato según la forma de onda moduladora.

```
SynthDef(\fm, {|gate=1 |
    var modulante, portadora, env;
    modulante=SinOsc.kr(4, 0, 100, 300);
    portadora=SinOsc.ar(modulante, 0, 1);
    env=EnvGen.kr(Env.asr(0.1, 1, 1), gate, doneAction:2);
    Out.ar(0, portadora*env)
    }).add
```

```
c=Synth(\fm)
c.set(\gate,0)
```

Modulación de Fase

Es un tipo de modulación FM, donde la fase de la frecuencia portadora es modulada por una frecuencia modulante.

```
(
SynthDef(\fase,{|gate=1,frec=300, frec_mod=2 |
  var modulante, portadora, env;
  modulante=SinOsc.kr(frec_mod,0,pi,pi);
  portadora=SinOsc.ar(frec,modulante,1);
  env=EnvGen.kr(Env.asr(0.1,1,1),gate, doneAction:2);
  Out.ar(0,portadora*env)
  }).send(s)
)
```

```
f=Synth(\fase)
f.set(\frec_mod,77)
f.set(\gate,0)
```

Bibliografía

Dodge, Ch. y A. Jerse, T. (1997). *Computer Music: Synthesis, composition and performance*. Schirmer.

Hutchins, C. (2005). *SuperCollider Tutorial: chapter 6*.

Valle, A. (2008). *The SuperCollider Italian Manual at CIRMA*. Torino: CIRMA.



Esta obra está sujeta a la licencia Attribution-NonCommercial-ShareAlike 3.0 Unported de Creative Commons. Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/3.0/> o envíe una carta a Creative Commons, 444 Castro Street, Suite 900, Mountain View, California, 94041, USA.